# *CLDP 948 Example 6b*

*Lesa Hoffman and Jihong Zhang*

*2018-10-29*

## 1 *Nominal Response Model (NRM):*

- Nominal Response Models (e.g. Bock, 1972) are models for ploytomous data where item responses are not ordinal.
- Information gained from the use of such models can be useful for detecting which distracting options are better than others in multiple choice.

### 1.1 *A Nominal Response Item*

Which political party would you identify yourself with ?

1. Democrat[1]
2. Republican
3. Independent
4. Green
5. Unaffiliated

[1] this option is choosen as anchor option

- The NRM specifies the likelihood that an examiner of a given ability will select option $k_j$ of item i.

### 1.2 *Parameters*

- Similar to GRM or PCM, NRM create a set of dummy variables to get probability of each category.
- Estimate k-1 slope parameter ($a_{ki}$) per item and k-1 difficulty parameter ($b_{ki}$) or intercepts ($d_{ki}$) (i.e. 5 options -> 4 difficulties and 4 discrimination), but the interpretation in NRM is very different from in PCM or GRM.

### 1.3 *Two parameterization methods*

- Bock, 1972

  $\sum a_k = \sum b_k = 0$

- Mellenbergh, 1995

  $a_1 = b_1 = 0$

### 1.4 Mirt example: NRM for 4-Category Item

`mirt` package provides convenient tools to fit NRM model. But the identification method is different from Mplus by Bock 1972. `mirt` uses the form $z = d + a_1 * ak * \theta$ where the ak terms serve as 'scoring' coefs (indicate the ordering of the categories) while the a1 represents the overall slope(discrimination) parameter across all categories. For d parameter, those are the likelihood of being endorsed when $\theta = 0$ (larger d values indicate that a category was more likely selected when theta equals 0).

The probability of choosing k category response is:

$$P(x = k|\theta) = \frac{e^{ak_{k-1}*(a_1*\theta)+d_{k-1}}}{\sum_1^k e^{ak_{k-1}*(a_1*\theta)+d_{(k-1)}}}$$

$$z_{k-1} = ak_{k-1} * a_1 * \theta + d_{k-1}$$

Then,

$$P(x = k|\theta) = \frac{e^z}{\sum_1^k e^z}$$

For example, for people whose trait level is 0, if d2 larger than d3, z2 will then be larger than z3. Then, the $e^{z_2}$ (nominator) will be larger than $e^{z_3}$. Since the denominator($\sum_1^k e^z$)) is same for all categories, the likelihood of choosing category 3 ($P_{x=3}$) will be larger than the likelihood of choosing category 4 (see Figure 1).

Table 1: NRM parameters

| | a1 | ak0 | ak1 | ak2 | ak3 | d0 | d1 | d2 | d3 |
|---|---|---|---|---|---|---|---|---|---|
| Comfort | 0.087 | 0 | -7.858 | -6.481 | 3 | 0 | 1.881 | 4.025 | 2.760 |
| Environment | 0.067 | 0 | -22.738 | -21.050 | 3 | 0 | 1.195 | 1.723 | 1.360 |
| Work | 0.155 | 0 | -5.940 | -3.637 | 3 | 0 | 1.089 | 1.918 | 0.237 |
| Future | 0.184 | 0 | -3.783 | -3.583 | 3 | 0 | 1.642 | 2.722 | 1.647 |
| Technology | 0.046 | 0 | -29.292 | -27.889 | 3 | 0 | 1.740 | 2.311 | 1.858 |
| Industry | 0.087 | 0 | -21.854 | -24.599 | 3 | 0 | 1.703 | 2.881 | 2.612 |
| Benefit | 0.137 | 0 | -6.477 | -5.188 | 3 | 0 | 1.601 | 2.303 | 1.072 |

### 1.5 Another Parameterization Method

Bock's (1972) original formulation of the nominal model was

$$P_{(u=k|\theta)} = \frac{exp(a_k\theta + c_k)}{\Sigma_i exp(a_k\theta + c_k)}$$

the curve tracing the probability that the item response u is in category k is a function of the latent variable $\theta$ with vector parameters a and c. k = 0,1,...,m − 1 for an item with m response categories.

To be identified, the sum of a1-a4 is constrained to 0 and the sum of c1-c4 is also constrained to 0. a is the slope parameters and c is the intercepts for each category.

Table 2: IRT version

| | a1 | a2 | a3 | a4 | c1 | c2 | c3 | c4 |
|---|---|---|---|---|---|---|---|---|
| Comfort | 0.2479192 | -0.4393110 | -0.3188852 | 0.5102770 | -2.1664458 | -0.2855718 | 1.8583455 | 0.5936722 |
| Environment | 0.6834829 | -0.8405704 | -0.7274799 | 0.8845673 | -1.0695872 | 0.1254844 | 0.6538681 | 0.2902346 |
| Work | 0.2551751 | -0.6666461 | -0.3092425 | 0.7207136 | -0.8110968 | 0.2781844 | 1.1068668 | -0.5739544 |
| Future | 0.2005983 | -0.4946839 | -0.4579307 | 0.7520163 | -1.5027516 | 0.1392060 | 1.2190123 | 0.1445333 |
| Technology | 0.6254204 | -0.7270765 | -0.6622835 | 0.7639396 | -1.4773630 | 0.2628401 | 0.8337212 | 0.3808017 |
| Industry | 0.9480691 | -0.9591746 | -1.1987801 | 1.2098856 | -1.7992223 | -0.0961823 | 1.0822432 | 0.8131614 |
| Benefit | 0.2973991 | -0.5917818 | -0.4148605 | 0.7092432 | -1.2440319 | 0.3570904 | 1.0589349 | -0.1719934 |




Figure 1: Category Response Curves

The author of mirt package said that in the nominal model this parametrization helps to identify the empirical ordering of the categories by inspecting the ak values. Larger values indicate that the item category is more positively related to the latent trait(s) being measured. For instance, if an item was truly ordinal (such as a Likert scale), and had 4 response categories, we would expect to see $a_{k_0} < a_{k_1} < a_{k_2} < a_{k_3}$ following estimation. If on the other hand $a_{k_0} > a_{k_1}$ then it would appear that the second category is less related to to the trait than the first, and therefore the second category should be understood as the 'lowest score'.

### 1.6 Mplus Version

Same data was estimated in NRM with Mplus. But parameterization is different from mirt results above.

Table 3: Parameter table mplus version

| item | Load1 | Load2 | Load3 | Int1 | Int2 | Int3 |
|---|---|---|---|---|---|---|
| Comfort | 0.243 | 0.953 | 0.896 | -2.758 | -0.843 | 1.286 |
| Environment | -0.216 | 1.559 | 1.477 | -1.634 | -0.220 | 0.293 |
| Work | 0.680 | 1.599 | 1.211 | -0.096 | 0.963 | 1.804 |
| Future | 0.647 | 1.449 | 1.375 | -1.576 | 0.057 | 1.149 |
| Technology | -0.452 | 1.254 | 1.262 | -2.283 | -0.176 | 0.367 |
| Industry | -0.386 | 1.846 | 2.097 | -3.089 | -0.980 | 0.201 |
| Benefit | 0.517 | 1.465 | 1.256 | -1.001 | 0.589 | 1.302 |

# 2 Semiordered Partical Credit Model

## 2.1 Example Data

This original data comes from the a 36-item stress measure that was developed to assess acculturative stress among persons of Mexican origin living in the United States, was tested on a community sample of 174 adults (117 women, 57 men). Seven items[2] were selected as sample data.

[2] the scale of items includes 5-point likert as well as **NA** option

1. CS1. It bothers me when people pressure me to assimilate to the American ways of doing things.
2. CS2. It bothers me when people do not respect my Mexican/Latino values.
3. CS3. Because of my cultural background, I have a hard time fitting in with Whites.
4. CS4. I feel uncomfortable when others expect me to know American ways of doing things.
5. CS5. I do not feel accepted by Whites.
6. CS6. I feel uncomfortable when I have to choose between Mexican/Latino and American ways of doing things.
7. CS7. People look down upon me if I practice Mexican/Latino customs.

Table 4: Item Scales

| | Does not apply | Not at all stressful | A little stressful | Somewhat stressful | Quite stressful | Extremely stressful |
|---|---|---|---|---|---|---|
| Original | 1 | 2 | 3 | 4 | 5 | 6 |
| Reversed | 6 | 5 | 4 | 3 | 2 | 1 |

**Mplus Code to Read in Data:**

```
TITLE:      GPCM via NRM for checking code
            Divide by loadings in steps only
DATA:       FILE = lesa.csv;
VARIABLE:
    ! cso = Original: 1=NA, then 2-6 Likert
    ! lik = Likert only: NA=missing, then 2-6
    ! nar = NA binary indicator: 0=NA, 1=Likert
    ! csr = Reverse: 6=NA, 5-1 Likert
    ! likr = Reverse: NA=missing, then 6-2
      !needed so that first=reference as in PCM because last=ref in NRM
    NAMES = ID cso1-cso11 lik1-lik11 nar1-nar11
            csr1-csr11 likr1-likr11;
    USEVARIABLES = csr1-csr7;
    NOMINAL = csr1-csr7;
```




Figure 2: Distributions of item response.

```
  MISSING = ALL (-99);

ANALYSIS:   ESTIMATOR = ML;
!PLOT:      TYPE = PLOT1 PLOT2 PLOT3;
!SAVEDATA:  SAVE = FSCORES; FILE = theta.dat; MISSFLAG=99;
```

---

## *2.2 Parameters*

When fitting nominal response models in Mplus, a set of dummy variables should be created. For example, a 6-category item will be divided to 5 dummy variables. Five sub-models are then estimated for this item:

- **6 vs. 1**: $\text{Logit}(6 \text{ vs. } 1) = -\tau_{5i} + \lambda_{5i} F_s$
- **5 vs. 1**: $\text{Logit}(5 \text{ vs. } 1) = -\tau_{4i} + \lambda_{4i} F_s$
- **4 vs. 1**: $\text{Logit}(4 \text{ vs. } 1) = -\tau_{3i} + \lambda_{3i} F_s$
- **3 vs. 1**: $\text{Logit}(3 \text{ vs. } 1) = -\tau_{2i} + \lambda_{2i} F_s$
- **2 vs. 1**: $\text{Logit}(2 \text{ vs. } 1) = -\tau_{1i} + \lambda_{1i} F_s$

Loadings: Load5-Load; Intercepts: Int5-Int1; Steps: Step4-Step1; Category: Cat5-Cat1

| | | | | | | | | | | |
|---|---|---|---|---|---|---|---|---|---|---|
| NA | | Not at all | | A little | | Somewhat | | Quite | | Extremely |
| 6 | | 5 | | 4 | | 3 | | 2 | | 1 |
| | Load5 | | Load4 | | Load3 | | Load2 | | Load | |
| | Int5 | | Int4 | | Int3 | | Int2 | | Int1 | |
| | | | Step4 | | Step3 | | Step2 | | Step1 | |
| | | | threshold4 | | threshold3 | | threshold2 | | threshold1 | |
| | | Cat5 | | Cat4 | | Cat3 | | Cat2 | | Cat1 |

Factor Loading [3] and item intercepts are reverse labeled to reverse the reverse-ordered item responses.

Model is identified with Z-Scored method (Factor Variance=1, Factor Mean = 0, each item has 5 estimated factor loading and intercepts). **Load_I** is the slope for 1 vs 2, which is set as reference.

[3] For example, **Load5_I1** refers to the loading of response 6 (Not Applicable) vs 5 (Not at all Stressful) for item 1; **Load4_I1** refers to the loading of original response 3 (Somewhat stressful) vs 4 (A little Stressful) for item 1; **Load3_I7** refers to the loading of 3 (Somewhat stressful) vs 4 (A little stressful) for item 7.

---

```
MODEL:
! GPCM is a special case of the NRM with lamda equals k-1
! Factor is presssure to acculturate (5 ordered categories, 1 nominal)
! Labels are backwards because to reverse the reverse-ordering
  PRESS BY
  csr1#1* csr1#2* csr1#3* csr1#4* csr1#5* (Load5_I1 Load4_I1 Load3_I1 Load2_I1 Load_I1)
  csr2#1* csr2#2* csr2#3* csr2#4* csr2#5* (Load5_I2 Load4_I2 Load3_I2 Load2_I2 Load_I2)
  csr3#1* csr3#2* csr3#3* csr3#4* csr3#5* (Load5_I3 Load4_I3 Load3_I3 Load2_I3 Load_I3)
  csr4#1* csr4#2* csr4#3* csr4#4* csr4#5* (Load5_I4 Load4_I4 Load3_I4 Load2_I4 Load_I4)
  csr5#1* csr5#2* csr5#3* csr5#4* csr5#5* (Load5_I5 Load4_I5 Load3_I5 Load2_I5 Load_I5)
  csr6#1* csr6#2* csr6#3* csr6#4* csr6#5* (Load5_I6 Load4_I6 Load3_I6 Load2_I6 Load_I6)
  csr7#1* csr7#2* csr7#3* csr7#4* csr7#5* (Load5_I7 Load4_I7 Load3_I7 Load2_I7 Load_I7);
```

```
! Label all submodel intercepts to reverse the reverse order
  [csr1#1-csr7#1*] (Int5_I1-Int5_I7); ! Will not be constrained
  [csr1#2-csr7#2*] (Int4_I1-Int4_I7); ! 5 vs 1
  [csr1#3-csr7#3*] (Int3_I1-Int3_I7); ! 4 vs 1
  [csr1#4-csr7#4*] (Int2_I1-Int2_I7); ! 3 vs 1
  [csr1#5-csr7#5*] (Int1_I1-Int1_I7); ! 2 vs 1
! Will become factor mean=0 and variance=1 for identification and labeled
  [PRESS*] (FactMean); PRESS* (FactVar);
```

---

**Model Constrain:**

*For Likert Responses:*

The semi-ordered GPCM is modeled as a constrained version of the NRM, the slopes[4] for the response categories are constrained such that

$$\lambda_{ik} = (k-1)\lambda_{ii}, k = 1, ..., K$$

in which $\lambda_{1i}$ is the slope for 2 vs. 1 dummuy variable in the GPCM. However, the slope for `NA vs. 1` sub-model[5] is not constrained and freely estimated.

The slopes for `k+1 vs. 1` sub-model are constrained to k*$\lambda_{1j}$ :

$$\lambda_{2i} = 2 * \lambda_{1i}$$

$$\lambda_{3i} = 3 * \lambda_{1i}$$

$$\lambda_{4i} = 4 * \lambda_{1i}$$

The step parameters [6] are the trait level needed for a 50% probability of endorsing the higher binary category.

$$s_{ik} = \frac{-(\tau_{ik} - \tau_{i(k-1)})}{\lambda_{ik}}$$

in which $\tau_{ik}$ [7] is the intercept for dummy coding sub-model of category k vs. 1.

*IRT version parameters:*

The slopes for likert responses and NA response are:

$$a_{ki} = \lambda_{1i}$$

$$a_{iNA} = \lambda_{5i}$$

Thus, the difficulty of NA category[8] could be calculated as:

$$b_{iNA} = -\frac{\tau_{iNA}}{\lambda_{iNA}} = -\frac{\tau_{5i}}{\lambda_{5i}}$$

---

[4] labeled as `Load(1,2,3,4)_I#` in *Mplus* code

[5] labeled as **Load5_I#** in Mplus

[6] labeled as **Step1_I#** in *Mplus*

[7] labeled as **Int1_I#** in *Mplus code*

[8] labeled as **BNa_I#** in *Mplus*

```
MODEL CONSTRAINT: ! Identification here so can use below
FactMean=0; FactVar=1;
! BELOW: DO (begin, end), replace # with index

! Slope for 1 vs 2 (Load) is estimated, others constrained as multiples
! Slope for NA category (Load5) is not constrained
  DO (1,7) Load4_I# = 4*Load_I#;
  DO (1,7) Load3_I# = 3*Load_I#;
  DO (1,7) Load2_I# = 2*Load_I#;

! Create new IRT parameters
! Define shared Likert and NA discriminations for convenience
  NEW(ALik_I1-ALik_I7 ANa_I1-ANa_I7);
  DO (1,7) ALik_I# = Load_I#;
  DO (1,7) ANa_I# = Load5_I#;

! Define difficulty for NA response from intercept
  NEW(BNa_I1-BNa_I7);
  DO (1,7) BNa_I# = -1*Int5_I#/Load5_I#;

! Define step difficulties (-1 because uses intercepts, not thresholds)
  NEW(Step1_I1-Step1_I7 Step2_I1-Step2_I7 Step3_I1-Step3_I7 Step4_I1-Step4_I7);
  DO (1,7) Step1_I# = -1*(Int1_I#-0)      /Load_I#; ! From 1 to 2
  DO (1,7) Step2_I# = -1*(Int2_I#-Int1_I#)/Load_I#; ! From 2 to 3
  DO (1,7) Step3_I# = -1*(Int3_I#-Int2_I#)/Load_I#; ! From 3 to 4
  DO (1,7) Step4_I# = -1*(Int4_I#-Int3_I#)/Load_I#; ! From 4 to 5

! Define average item locations
  NEW (Loc_I1-Loc_I7);
  DO (1,7) Loc_I# = (Step1_I#+ Step2_I#+ Step3_I#+ Step4_I#)/4;

! Define "item categories"
  NEW(Cat1_I1-Cat1_I7 Cat2_I1-Cat2_I7 Cat3_I1-Cat3_I7 Cat4_I1-Cat4_I7 Cat5_I1-Cat5_I7);
  DO (1,7) Cat1_I# = 0;
  DO (1,7) Cat2_I# = -1*(Step1_I#-Loc_I#);
  DO (1,7) Cat3_I# = -1*(Step2_I#-Loc_I#);
  DO (1,7) Cat4_I# = -1*(Step3_I#-Loc_I#);
  DO (1,7) Cat5_I# = -1*(Step4_I#-Loc_I#);

! Recreate Mplus thresholds labeled "steps"
  NEW(T1_I1-T1_I7 T2_I1-T2_I7 T3_I1-T3_I7 T4_I1-T4_I7);
  DO (1,7) T1_I# = Load_I#*Step1_I#;
  DO (1,7) T2_I# = Load_I#*Step2_I#;
```

```
DO (1,7) T3_I# = Load_I#*Step3_I#;
DO (1,7) T4_I# = Load_I#*Step4_I#;
```

---

## 2.3 Output

Table 5: Predicted Probability for CSR1

| | Category 1 Extremely | Category 2 Quite | Category 3 Somewhat | Category 4 A little | Category 5 Not at all | Category 6 NA |
|---|---|---|---|---|---|---|
| $\theta$ | $P_1(\theta)$ | $P_2(\theta)$ | $P_3(\theta)$ | $P_4(\theta)$ | $P_5(\theta)$ | $P_6(\theta)$ |
| -3.0 | 0.8879 | 0.0932 | 0.0173 | 0.0015 | 0.0001 | 0.0000 |
| -2.5 | 0.8424 | 0.1222 | 0.0313 | 0.0038 | 0.0003 | 0.0000 |
| -2.0 | 0.7785 | 0.1559 | 0.0552 | 0.0093 | 0.0010 | 0.0001 |
| -1.5 | 0.6904 | 0.1910 | 0.0933 | 0.0216 | 0.0031 | 0.0005 |
| -1.0 | 0.5741 | 0.2194 | 0.1479 | 0.0474 | 0.0095 | 0.0017 |
| -0.5 | 0.4330 | 0.2285 | 0.2128 | 0.0942 | 0.0261 | 0.0055 |
| 0.0 | 0.2848 | 0.2076 | 0.2669 | 0.1632 | 0.0624 | 0.0151 |
| 0.5 | 0.1581 | 0.1592 | 0.2826 | 0.2386 | 0.1261 | 0.0353 |
| 1.0 | 0.0735 | 0.1022 | 0.2503 | 0.2920 | 0.2131 | 0.0690 |
| 1.5 | 0.0291 | 0.0559 | 0.1889 | 0.3044 | 0.3068 | 0.1149 |
| 2.0 | 0.0101 | 0.0269 | 0.1256 | 0.2796 | 0.3892 | 0.1685 |
| 2.5 | 0.0032 | 0.0118 | 0.0761 | 0.2339 | 0.4498 | 0.2252 |
| 3.0 | 0.0010 | 0.0048 | 0.0431 | 0.1831 | 0.4865 | 0.2815 |

Table 6: Likert Responses

| | **Slope** | | **Step1 (1 vs 2)** | | **Step2 (2 vs 3)** | | **Step3 (3 vs 4)** | | **Step4 (4 vs 5)** | |
|---|---|---|---|---|---|---|---|---|---|---|
| *Item* | *Estimate* | *SE* | *Estimate* | *SE* | *Estimate* | *SE* | *Estimate* | *SE* | *Estimate* | *SE* |
| 1 | 0.646 | 0.095 | 0.489 | 0.322 | -0.388 | 0.303 | 0.762 | 0.302 | 1.488 | 0.373 |
| 2 | 0.528 | 0.090 | -0.239 | 0.499 | -1.601 | 0.478 | -0.330 | 0.369 | 0.005 | 0.353 |
| 3 | 1.878 | 0.294 | 0.087 | 0.113 | 0.659 | 0.126 | 1.153 | 0.161 | 1.684 | 0.227 |
| 4 | 1.623 | 0.255 | -0.195 | 0.124 | 0.541 | 0.142 | 0.688 | 0.155 | 1.407 | 0.200 |
| 5 | 1.932 | 0.315 | 0.349 | 0.116 | 0.849 | 0.133 | 1.433 | 0.190 | 1.732 | 0.246 |
| 6 | 0.881 | 0.123 | 0.243 | 0.211 | 0.383 | 0.216 | 1.044 | 0.251 | 2.093 | 0.373 |
| 7 | 1.309 | 0.190 | 0.281 | 0.143 | 1.034 | 0.185 | 1.133 | 0.220 | 1.860 | 0.305 |

## 2.4 Figures




Figure 3: Category Response Curves

Table 7: NA Category

| | Discrimination (NA vs 1) | | Difficulty | |
|---|---|---|---|---|
| *Item* | *Estimate* | *SE* | *Estimate* | *SE* |
| 1 | 2.873 | 0.583 | 1.022 | 0.171 |
| 2 | 2.666 | 0.396 | -0.237 | 0.130 |
| 3 | 8.893 | 1.697 | 1.104 | 0.139 |
| 4 | 7.860 | 1.298 | 0.840 | 0.118 |
| 5 | 5.824 | 1.745 | 1.127 | 0.146 |
| 6 | 1.311 | 0.620 | 1.930 | 0.904 |
| 7 | 5.668 | 1.324 | 1.365 | 0.171 |




Figure 4: Item Information Curves




Figure 5: Test Information Curve




Figure 6: Reliability Curves